2.4. Лабораторная работа 4. Обработка текстовой информации
Цель работы:

· освоение операций со строковыми данными на Паскале;
· владение методикой обработки строковых данных.

2.4.1. Теоретические сведения, необходимые для выполнения работы
Тип данных String
Для удобства обработки текстовой информации на языке Паскаль предусмотрен специальный тип String (строка). Он позволяет эффективно работать с цепочкой символов длиной не более 255. Переменная типа String может быть объявлена в одной из двух форм:

Имя: string;

Имя: string [константное выражение];

Константное выражение в квадратных скобках указывает максимально возможную длину строковой переменной. Если этот атрибут отсутствует в описании, то по умолчанию он равен 255. Если указано string[10], это означает, что переменная имеет длину не более 10 символов. Особенность переменных типа string в том, что они занимают в памяти на 1 байт больше объявленной длины строки. Нулевой байт этой области называют байтом текущей длины. В нем хранится двоичное число в формате byte. Если в нулевом байте содержится "0", то строка пустая.

Пример 1. Описание строковых переменных.

Var X: string [5];

    Name: string;

Переменная X занимает в памяти 6 байт и может иметь длину от 0 до 5, переменная Name – 256 байт, может иметь длину от 0 до 255. 

Пример 2. Операторы присваивания для переменной X, описанной выше.

X:='777';      {x – строка из 3 знаков}

X:=' ';        {x – строка из одного знака пробела}

X:='';         {x – пустая строка, ее текущая длина равна 0}

X:='abcdefg';  {x – строка из 5 знаков 'abcde'}

Переменная типа string может быть элементом списка процедуры ввода Readln. Например, ввод строковой переменной X: string[5]:

writeLn ('ввести строку Х'); ReadLn(X);

Если набрано больше 5 символов, то из буфера ввода в поле переменной Х будут скопированы только первые 5. Если ничего не набрано и нажата клавиша Enter, то в поле Х в нулевой байт занесется число 0, то есть переменная Х имеет значение пустой строки.

Строковые данные могут быть напечатаны с помощью процедуры write или writeln. 

Пример 3. Печать строковых переменных.

write (X); {ширина поля вывода равна текущей длине}

X:='ABC'; 
write (X:2); {символ 'C' не будет выведен}

Строковые данные можно выводить посимвольно. Например, печать строки по одному знаку: x := 'abcde': For i := 1 to 5 do write(x[i]);

Сравнение строк

Для сравнения строк используются знаки операций: < , <= , = , >= , > , <> .

Строки сравниваются слева направо посимвольно с учетом ASCII кодировки символов в соответствующих позициях. Нулевые строки могут быть равны только другим нулевым строкам. При сравнении строк разной длины недостающие символы короткой строки считаются всегда меньше любого действительного символа в длинной. Например, строка 'a' меньше строки 'a1'. Сначала сравниваются самые левые символы; если они равны, то сравниваются следующие и т. д. Например, строка 'AS9' больше строки 'AS6'.

Допустимы массивы строк: SPISOK: array [1..5] of string [20];
Каждый элемент массива строк будет занимать одинаковое количество байт, но текущая длина каждого элемента может быть различна.

Пример 4. Программа, которая читает массив строк, а затем печатает строки в алфавитном порядке. 
Замечание: среди вводимых строк не должно быть пустых. Программа просматривает массив строк, не принимая во внимание пустые строки, находит и печатает строку с наименьшей длиной. В позицию наименьшей строки заносится пустая строка. Программа заканчивает работу, когда не остается строк для печати.
Program sort_print;

Const size=10;

Var

 I,j,k: integer;

 List:array[1..size] of string;

Begin

 Writeln(' Введите ',size,' строк текста');

 For i:=1 to size do begin write(i,' -я строка');readln(list[i]); end;

 { печать по алфавиту}

 writeln;

 for i:=1 to size do

 begin

  j:=0; repeat j:=j+1 until list[j]<>'';

  for k:=j+1 to size do

   if (list[k]<list[j]) and( list[k]<>'') then k:=j;

  writeln(list[k]);

  list[j]:='';

 end;

 readln;

End.

Пример 5. Решение предыдущей задачи с помощью сортировки массива строк.

Program sort_string;

Const  size=10;

Var

 I,j,k: integer;

 List:array[1..size] of string;

 Temp:string; {рабочая строка для перестановки элементов массива}

Begin

 Writeln(' Введите ',size,' строк текста');

 For i:=1 to size do begin write(i,' -я строка');readln(list[i]); end;

 {Сортировка элементов в алфавитном порядке}

 writeln;

 for i:=1 to size-1 do

  for k:=i+1 to size do

  if list[i]>list[k] then 

   begin {перестановка строк i-й -и k-й}

    temp:=list[i]; list[i]:=list[k]; list[k]:=temp

   end;

 writeln; {печать списка строк в алфавитном порядке}

 for i:=1 to size do writeln(list[i]);

 readln;

End.

Стандартные процедуры и функции для обработки строк

В Паскале определены следующие стандартные подпрограммы обработки строк:

а) Функции Cоncat, Length, Copy, Pos.
б) Процедуры Delete, Insert, Str, Val.

Функция Cоncat выполняет соединение (конкатенацию) списка строк.

Concat(s1, [s2, ..., sn]; string)

Тип результата: строковый (string). Если длина результирующей строки превышает 255 символов, то она обрезается до 255 символов. Функция аналогична операции "+".

Функция Сору возвращает подстроку строки.

Соpy(s, index, count)

Тип результата: строковый (string).

Параметр s представляет собой выражение строкового типа. Параметры index и count являются выражениями целого типа.

Функция Length возвращает динамическую длину строки.

Length(s)

Тип результата: целый (integer).

Параметр s представляет собой выражение строкового типа. Результатом будет длина строки s.

Функция Роs производит поиск подстроки в строке.

Pos(st1, st2)

Тип результата: байт (byte).

Параметры st1 и st2 являются выражениями строкового типа. Данная функция ищет подстроку, заданную параметром st1 в строке st2, и возвращает целое значение, являющееся позицией первого символа подстроки в строке s. Если подстрока не найдена, то функция возвращает значение 0.

Пример 6. Демонстрация работы функции pos.
Program dem_pos;

var

 s: string;

begin

 s := '    123.5';

 { преобразует пробелы в нули }

 while Pos(' ',s) > 0 do s[Pos(' ',s)] := '0';

end.

Процедура Insert(scr, res, index) вставляет строку scr внутрь res, начиная с позиции index. Параметр scr представляет собой выражение строкового типа. Параметр res – переменная строкового типа любой длины. Параметр index является выражением целого типа. Если получившаяся в результате строка превышает 255 символов, то она усекается до 255 символов.

Пример 7. Демонстрация работы процедуры insert.

Program dem_insert;

var

 s: string;

begin

 s := 'Honest Lincoln';

 Insert('Abe',s,8); { Honest Abe Lincoln }

...

Процедура Delete(s, index, count) удаляет из строки s подстроку длиной count, начиная с позиции index. Если значение параметра index превышает длину строки, то символы не удаляются. Если параметр count задает больше символов, чем остается в строке, начиная c позиции index, то удаляется остаток строки.

Пример 8. Демонстрация работы процедуры Delete.

Program dem_delete;

Var

 S: String;

Begin

 S:= 'Television';

 Delete(s,1,4); {Vision}

...

Процедура Str(Val_n, Var_s) преобразует численное значение Val_n в его строковое представление, которое присваивает переменной Var_s. Val – это арифметическое выражение, а Var_s – имя строковой переменной. После выражения Val_n может следовать выражение целого типа, которое задает количество символов в Var_s.

Пример 9. Демонстрация работы процедуры Str.

Program dem_str;

Var b:string;

    k:real;

begin

 k:=-2.78465;

 str(k:5:2,b); {пpеобpазование числа в стpоку b='-2.78'}

...

Процедура Val(s, v, code) – преобразует строковое значение s в число и присваивает его значение переменной v. Параметр v является переменной целого или вещественного типа. Параметр code – это переменная целого типа, которая хранит код завершения. Если преобразование прошло благополучно, то значение code=0, иначе code содержит номер первого ошибочного символа в строке s.

Пример 10. Демонстрация работы процедуры val.

Program dem_val;

Var b:string;

    k:real;

    i:integer;

begin

 b:='-12.456';

 val(b,k,i);

 if i<>0 then write('Hедопустимый символ в стpоке')

 else begin {ноpмальное пpеобpазование}

        k:=k*2;

        write(k)

      end;

end.

Пример 11. Удалить пробелы в начале текста и лишние пробелы между словами. Сформировать массив из слов текста.

Program prim_lab;

Const Doubl='  '; {2 пробела}

VAR 

 S:string;

 i,k:byte;

 Massiv:array[1..100] of string;

Begin

 Write('ввод строки текста'); readln(s);

 {Удалим пробелы слева} While s[1]=’’ do delete(s,1,1);

 s:=s+' '; {обеспечим пробел в конце текста}

 repeat

  i:=pos(doubl,s); {i – позиция двух пробелов в тексте}

  if i>0 then delete(s,i,1) {удаление первого пробела из пары}

 until i=0;

 {Переносим выделенные слова в массив, после чего они удаляются из исходного текста}

 k:=0; {счетчик слов в формируемом массиве}

 repeat

  i:=pos(‘ ‘,s);

  if i>0 then

  begin

   inc(k);
   {запись в massiv найденного слова}

   massiv[k]:=copy(s,1,i-1);
   delete(s,1,i) {удаление первого слова c пробелом}

  end

 until s=’’; {повторять, пока строка не пуста}

 {вывод массива слов}

 for i:=1 to k do writeln(massiv[i]);

end.

2.4.2. Задания к лабораторной работе
Общая постановка задачи для всех вариантов

Задан текст, состоящий из слов, записанных через  пробелы или запятые. Длина текста не больше 255 символов. В тексте могут быть использованы любые символы. Словом считать последовательность символов, не содержащую пробел. Необходимо введенный текст вывести на экран до и после его преобразования. 

Индивидуальные задания на строки

13. Сформировать новую строку, в которой слова будут следовать в обратном порядке. Найти первое цифровое слово и заменить его обратным значением.
2.4.3. Контрольные вопросы и задания

1. Два способа представления текста в Паскале.

2. Тип данных string: описание переменных, атрибут длины, ввод, вывод и операция присваивания. Приведите примеры.

3. Сравнение строк.

4. Стандартные процедуры перевода строки в число и числа в строку. Приведите примеры.
5. Назначение стандартных функций Length(s), Pos(s), Copy(s).
2.5. Лабораторная работа 5. Использование процедур и функций

2.5.1. Теоретические сведения, необходимые для выполнения работы
Модульное программирование

Программу, состоящую только из большого числа операторов, называют монолитной. До сих пор мы писали только такие программы.

Модульное программирование – это организация программы как совокупности небольших независимых блоков, называемых модулями.
Принцип “разделяй и властвуй” известен со времен Юлия Цезаря и является методом решения трудных проблем путем разделения проблемы на множество мелких независимых, которые легче понимать и решать. В разработке программ этот принцип означает разделение программы на отдельные фрагменты (модули), которые просты по управлению и допускают независимую отладку и тестирование, что позволяет разработчику программ спокойно работать над отдельной частью большой системы, не заботясь обо всех ее деталях.

Часто модули программы сами состоят из модулей более низкого уровня, то есть модульная программа имеет иерархическую структуру. Программист пишет основную (главную) программу и несколько подпрограмм. Для общности программу и подпрограммы называют модулями. Модуль имеет одну входную и одну выходную точку. Модуль, в котором записано обращение к другому модулю, называется вызывающим; второй модуль называется вызываемым. Основная программа – это модуль, который вызывается только операционной системой. Каждому модулю присваивается имя, и модульная структура программы изображается в виде иерархической схемы модулей (рис. 9).

Удачная структура модульной программы позволяет улучшить процесс ее разработки. Число модулей, вызываемых каким-либо модулем, называется размахом, или шириной управления модулями. Слишком большая ширина управления является признаком плохой схемы разбивки на модули. 

Среди множества модулей различают: головной модуль – управляет запуском программного продукта (существует в единственном числе); управляющий модуль – обеспечивает вызов других модулей на обработку; рабочие модули – выполняют функции обработки; сервисные модули и библиотеки – осуществляют обслуживающие функции.

В работе программного продукта активизируются необходимые программные модули. Управляющие модули задают последовательность вызова на выполнение очередного модуля. Информационная связь модулей обеспечивается либо через общие данные, либо межмодульной передачей данных через переменные обмена, называемые параметрами.

Структурный подход к программированию состоит из трёх частей: нисходящее проектирование, структурное программирование и тестирование.

Нисходящее проектирование

Разработку алгоритма удобно вести, двигаясь по дереву сверху вниз, как говорят, нисходящим методом, то есть от общего к частному. На первом этапе весь алгоритм представляется в виде единственного процедурного блока. Затем исходный алгоритм разбивается на части, выполняющие определенные функции. Процесс разработки алгоритма продолжается аналогично, пока весь алгоритм не будет разложен на достаточно простые блоки. Модульная структура программы позволяет сократить текст программы за счет выделения одинаковых участков обработки в отдельные модули. Разбивать программу на отдельные модули рекомендуется даже в тех случаях, когда эти модули используются однократно. На рис. 9. представлен пример схемы иерархии модулей.

Рис. 9. Пример иерархии модулей
Представление программы в виде иерархии относительно обособленных фрагментов со строго определёнными интерфейсами делает её легко проверяемой и приводит к повышению качества и эффективности программы.

Модуль или подпрограмма – это часть программы, оформленная в виде отдельной синтаксической конструкции и снабженная именем.

Любая подпрограмма кроме задания последовательности действий может содержать описание некоторой совокупности локальных объектов – констант, типов, переменных и так далее. Эти объекты предназначены для внутреннего использования внутри этой подпрограммы. 

Подпрограмма может быть предназначена для работы с различными данными. Гибкость и универсальность подпрограммного механизма обеспечивают с помощью понятия параметров. 

Общая структура подпрограммы

Структура подпрограммы подобна структуре паскаль-программы, что имеет глубокий смысл: «Часть подобна целому». 

При описании подпрограммы необходимо задать три основные компоненты:

· интерфейс подпрограммы, то есть информацию, необходимую для её вызова;

· совокупность описаний локальных рабочих объектов;

· собственно действия (операторы), составляющие смысл подпрограммы.

Интерфейс задается с помощью  заголовка. Вторая часть представляет собой раздел описаний локальных данных, третья – операторы вычислений. В дальнейшем будем называть блок описаний и операторов телом подпрограммы.

В языке Паскаль имеется два вида подпрограмм: процедуры и функции. 

Подпрограмму или модуль обычно оформляют в виде функции, если результатом ее выполнения является одно скалярное значение. Если подпрограмма в результате работы не выдает ни одного результата, например, выполняет графические построения или выдает больше одного результата, то ее оформляют в виде процедуры.

Например, видом модуля, выполняющего транспонирование матрицы, очевидно, будет процедура, а модуль, выполняющий вычисление определителя матрицы, следует оформить как функцию. 

Для досрочного выхода из подпрограммы используется стандартная процедура Exit. 

Процедуры и функции

Описания процедур и функций должны располагаться в программе после описания переменных перед секцией основного кода, который находится между последними операторами begin и end программы и управляет логикой программы.
Использование процедур и функций позволяет записать программу короче и наглядней. Все детали убраны в тела процедур и функций, и логика модульной программы достаточно проста для понимания.

Описание процедуры

Описание процедуры позволяет связать идентификатор с процедурным блоком:
<Описание процедуры> ::= <заголовок процедуры><тело процедуры>;

Заголовок именует идентификатор процедуры и задает список формальных параметров (если они имеются):
<заголовок процедуры> :: =procedure <идентификатор>;|

     procedure <идентификатор> (список формальных параметров);

<тело процедуры> ::= <раздел описаний> begin <операторы> end;|

                                     begin <операторы> end;

Операторы, которые выполняют назначение процедуры, содержатся в операторной части модуля процедуры между begin и end.

В качестве примера рассмотрим описание процедуры, выполняющей ввод с контролем ошибок ввода переменной вещественного типа. 

Идентификатор процедуры read_real. Входные параметры процедуры: X – адрес вводимой переменной и два значения X1 и X2, определяющие границы допустимых значений переменной X. 

Описание процедуры имеет вид
Procedure read_real(var x:real; x1,x2:real);{ Заголовок процедуры }

Var                              { раздел описаний}

    Iocode:integer;              { локальных      }

     Fl:boolean;                 { переменных     }

Begin
{   Раздел операторов процедуры    } 

 Repeat
 Write(' Введите число в интервале от  ', x1, ' до ',x2);

{$I-} readln(x);

{$I+} iocode:=ioresult;

  fl:= (iocode=0) and(x>=x1) and(x2<=x2);

  if not fl  then writeln(' Неверное значение! Повторите ввод');

until fl;

end; { конец описания процедуры read_real }
В начале работы процедуры отменяется автоматическая проверка ошибок ввода-вывода с помощью директивы компилятора {$I-}. Для ввода данного используется цикл repeat ... until до тех пор, пока не будет введено правильное значение Х.

Процедура активизируется с помощью оператора процедуры, в котором содержатся имя процедуры и необходимые параметры. Для ввода переменной z, значение которой лежит в интервале от -5.5 и 9, нужно записать вызов процедуры следующего вида:

 read_real(z,-5.5,9).
Директивы near и far

Для некоторых специальных целей может потребоваться использовать модель с дальним типом вызова. Чтобы переопределить автоматический выбор модели вызова компилятором, можно использовать директиву компилятора {$F+}. Процедуры и функции, компилируемые в состоянии {$F+}, всегда будут иметь дальний тип вызова far. По умолчанию используется директива {$F-}.

Директива forward

Описание процедуры, содержащее вместо блока операторов директиву forward, называется опережающим описанием.

Определяющее описание – это описание, в котором используется тот же идентификатор процедуры, но опущен список формальных параметров, и в которое включен блок операторов. Описание forward и определяющее описание должны присутствовать в одной и той же части описания процедуры и функции. Между ними могут описываться другие процедуры и функции, которые могут обращаться к процедуре с опережающим описанием.

Опережающее описание и определяющее описание представляют собой полное описание процедуры. 

Пример программы с опережающим описанием процедуры:

Program example;

 procedure two(m,n : integer); forward;

 procedure one(x,y : real);

  begin

   two(int(x),int(y));

  end;

procedure two;

begin

  writeln(m,n);
end;

begin

  one(-3.7, 10);

. . . 

end.

Описание функции

Описание функции похоже на описание процедуры, но имеются три различия:

1. Заголовок модуля-функции начинается с ключевого слова function и в отличие от процедуры заканчивается указанием типа возвращаемого результата.

function <идентификатор>(список формальных параметров):< тип>;
Функция не может возвращать структурный тип (массив, запись или файл).

2. В операторной части блока функции среди операторов, которые должны выполняться при активизации функции, должен быть по крайней мере один оператор присваивания, в котором идентификатору функции присваивается значение. Результатом функции является последнее присвоенное значение.

Вызов функции аналогичен обращению к встроенным функциям Паскаля, то есть указывается имя функции и в круглых скобках аргументы, необходимые для вычисления функции. 

Пример. Составить описание функции, находящей наибольший общий делитель двух натуральных чисел m и n. Имя функции будет Nod, на входе два числа m и n, меньшие 215 -1. Возвращаемое значение будет того же типа, что и входные данные.

Function Nod(m,n:word):word;

{   Алгоритм Евклида   }

Var
  d :   word; { рабочая локальная переменная }

Begin

 Repeat

     d:=m mod v;

     m:=n;

     n:=d;

Until n=0;

Nod:=m; { оператор присвоения идентификатору }

        { функции возвращаемого результата }

end;

Аналогично процедурам функции могут описываться как с ближним типом вызова (near), так и с дальним (far) и как опережающие (forward).

Локальные и глобальные определения данных

Переменные, описанные внутри подпрограммы, являются ее локальными переменными. Локальные переменные "невидимы" и недоступны вне подпрограммы.

Переменные, описанные в главной программе, называются глобальными. При входе в подпрограмму низшего уровня становятся доступными не только объявленные в ней переменные, но и сохраняется доступ ко всем переменным верхнего уровня. 

Исключение составляют те глобальные переменные, имена которых совпадают с именами переменных, локализованных в подпрограмме. В этом случае считается, что локальное имя "закрывает" глобальное и делает его недоступным, например:

Var      i:integer; {глобальная переменная}

Procedure P;

Var      i: integer; {локальная переменная}

Begin

 Writeln(i);

End;

Begin

 I:=1; P;

End;

Что напечатает данная программа? Значение локальной переменной i при входе в процедуру P не определено, хотя одноименная глобальная переменная имеет значение 1. Если убрать описание переменной i из процедуры, то на экран будет выдано значение глобальной переменной i, то есть 1.

Все вышесказанное относится не только к переменным, но и к типам, константам.

Параметры подпрограмм

В описании процедуры или функции задается список формальных параметров. Формальный параметр или фиктивный аргумент используется для обозначения входных (выходных) данных подпрограммы. Формальные параметры являются локальными по отношению к описываемой процедуре или функции.

В операторах тела подпрограммы все действия описаны с формальными параметрами. Но при активизации подпрограммы формальные параметры заменяются на фактические и вычисления ведутся с фактическим данными.

Для формальных параметров и локальных переменных подпрограмм на время выполнения подпрограммы выделяется место в оперативной памяти машины в соответствии с их типом и размерами. Эта область памяти носит название стек. В стек заносится либо значение фактического параметра, либо адрес его расположения в памяти.

Существует четыре вида параметров: значение, константа, переменная и нетипизированная переменная. Они характеризуются следующим:

1. Группа параметров без предшествующего ключевого слова является списком параметров-значений.

2. Группа параметров, перед которыми следует ключевое слово const и за которыми следует тип, является списком параметров-констант.
3. Группа параметров, перед которыми стоит ключевое слово var и за которыми следует тип, является списком типизированных параметров-переменных.

4. Группа параметров, перед которыми стоит ключевое слово var или const, за которыми не следует тип, является списком нетипизированных параметров.

Параметры-значения

Формальный параметр-значение обрабатывается как локальная по отношению к процедуре или функции переменная, за исключением того, что он получает свое начальное значение из соответствующего фактического параметра при активизации процедуры или функции. Изменения, которые претерпевает формальный параметр-значение, не влияют на значение фактического параметра.

Фактический параметр должен иметь тип, совместимый по присваиванию с типом формального параметра-значения и может быть выражением. Если параметр имеет строковый тип, то формальный параметр будет иметь атрибут размером 255 символов.

Параметры-константы

Формальные параметры-константы работают аналогично локальной переменной, доступной только по чтению, которая получает свое значение при активизации процедуры или функции от соответствующего фактического параметра. Присваивания формальному параметру-константе не допускаются. Формальный параметр-константа также не может передаваться в качестве фактического параметра другой процедуре или функции.

function Max(const a: Vector; n: integer): extended;

Параметр-константа, соответствующий фактическому параметру в операторе процедуры или функции, должен подчиняться тем же правилам, что и фактическое значение параметра. В тех случаях, когда формальный параметр не изменяет при выполнении процедуры или функции своего значения, вместо параметра-значения следует использовать параметр-константу.

Параметры-переменные

Параметр-переменная используется, когда значение должно передаваться из процедуры или функции вызывающей программе. Соответствующий фактический параметр в операторе вызова процедуры или функции должен быть ссылкой на переменную. При активизации процедуры или функции все действия подпрограммы выполняются над фактическими параметрами, любая ссылка на формальный параметр-переменную приводит к доступу к самому фактическому параметру .

Procedure NullsArray(var a: Vector; n: integer);

Var i:Integer;

Begin

 {обнуление массива}

 For i:=1 to n do a[i]:=0;

End;

Тип фактического параметра должен совпадать с типом формального параметра-переменной.

Примечание: файловый тип может передаваться только как параметр-переменная.

Нетипизированные параметры

Когда формальный параметр является нетипизированным параметром-переменной, то соответствующий фактический параметр может представлять собой любую ссылку на переменную или константу, независимо от ее типа. Нетипизированный параметр, описанный с ключевым словом var, может модифицироваться, а нетипизированный параметр, описанный с ключевым словом const, доступен только по чтению.

Нетипизированный параметр несовместим с переменными всех типов, пока ему не будет присвоен определенный тип с помощью присваивания типа переменной.
Процедурные типы и переменные. Параметры-процедуры и параметры-функции

Для объявления процедурного типа используется заголовок процедуры (функции) без указания имени процедуры или функции, например:

Type

  Proc1 = Procedure (a, b: real; var c: real);

  Proc2 = Procedure (var a, b: integer );

  Func1 = Function: String;

  Func2 = Function (a: Real): Real;

Имена параметров в описании процедурного типа никакого действия не вызывают. После определения процедурного типа появляется возможность описывать переменные этого типа. Такие переменные называют процедурными переменными. Например, с учетом описаний типа из предыдущего примера, можно объявить следующие переменные:

var

P: Proc2;

F: Func2;

Процедурной переменной можно присвоить значение только процедурного типа. Таким значением может быть другая процедурная переменная, или идентификатор процедуры или функции. Например, пусть мы имеем следующие описания процедуры и функции:

procedure Swap(var A,B: integer);

var Temp: integer;

begin

 Temp := A;

 A := B;

 B := Temp;

End;

function Tan(Angle: real): real;

begin

 Tan := Sin(Angle) / Cos(Angle);

end.

Описанным ранее переменным P и F теперь можно присвоить значения: 
P := Swap;
F := Tan;

После такого присваивания обращение P(i,j) эквивалентно Swap(i,j) и F(X) эквивалентно Tan(X).

Процедурные переменные в левой и в правой части должны быть совместимы по присваиванию, то есть должны иметь одно и то же число параметров, а параметры на соответствующих позициях должны быть одинакового типа. Нельзя присвоить процедурной переменной стандартную процедуру или функцию.

Когда процедурной переменной присваивается идентификатор процедуры, это означает, что адрес процедуры заносится в поле процедурной переменной. Формальные параметры подпрограмм могут иметь процедурный тип. Если процедура или функция должны передаваться в качестве параметра, они должны удовлетворять тем же правилам совместимости типа, что и при присваивании. То есть такие процедуры или функции должны компилироваться с директивой far, они не могут быть встроенными функциями и не могут быть вложенными.

Программирование и отладка модульных программ

При нисходящей разработке программирование модулей программы начинается с модуля самого верхнего уровня. Переход к программированию какого-либо другого модуля выполняется в случае, если уже запрограммирован модуль, который к нему обращается. 

Первым тестируется головной модуль программы, который представляет всю тестируемую программу. Те модули, к которым может обращаться головной, заменяются их имитаторами (так называемыми заглушками). Каждый имитатор модуля, в основном, сигнализирует о самом факте обращения к имитируемому модулю и выдает, если это необходимо, заранее запасенный подходящий результат. После завершения тестирования и отладки головного и любого последующего модуля производится переход к тестированию одного из модулей, которые в данный момент представлены имитаторами. Для этого имитатор выбранного для тестирования модуля заменяется самим этим модулем и кроме того добавляются имитаторы тех модулей, к которым может обращаться выбранный для тестирования модуль.

Пример применения модульного программирования

Рассмотрим решение следующей задачи. Пусть даны две квадратные матрицы С и Т. Если количество отрицательных элементов в матрице С больше, чем в матрице Т, то вычислить матрицу F=(C*T)T, иначе F=CT*T.

Данную задачу можно разбить на следующие подзадачи: ввод матрицы, подсчет количества отрицательных элементов матрицы, транспонирование матрицы, умножение двух матриц, печать матрицы. Решение каждой из них оформим  в виде отдельного модуля. 

Схема иерархии модулей будет иметь вид (рис. 10):

Рис. 10. Схема иерархии модулей задачи
Matrix_Work – главная программа. 

Приведем спецификации для каждого модуля (Табл. 6).
Спецификация модулей


Таблица 6
	№
	Имя
	Назначение
	Тип п/п
	Параметры
	Тип результата

	1
	Input
	Ввод квадратной матрицы
	Процедура
	1. А – квадратная матрица

2. m – порядок матрицы
	-

	2
	Koln
	Определение количества отрицательных элементов
	Функция
	1. А – квадратная матрица

2. m – порядок матрицы
	Целый

	3
	Umn
	Умножение двух матриц
	Процедура
	1. А – первая матрица (вх.)

2. B – вторая матрица (вх.)

3. m – порядок матриц (вх.)

4. r – результат А*В (вых.)
	-

	4
	Trans
	Транспонирова-ние матрицы
	Процедура
	1. B-квадратная матрица (вх.)

2. m – порядок матрицы

3. r – результата Bt (вых.)
	-

	5
	Print
	Печать матрицы
	Процедура
	1. А – квадратная матрица

2. m – порядок матрицы
	-


Блок-схема главной программы представлена на рис. 11. Прямоугольник с двойными боковыми стенками обозначает вызов подпрограммы.

Рис. 11. Блок схема главного модуля (программы) Matrix_work
Текст программы представлен ниже.

program Matrix_work;

uses crt;

const

 n=5 ; {максимальный порядок матрицы}

type

 matrix=array[1..n,1..n] of real;

var

 row : integer; {Порядок матриц}

 c,t : matrix;  {Исходные матрицы}

 f : matrix;    {Результирующая матрица}

 r,tt : matrix; {Рабочие матрицы}

 i,j : integer;

 kc,kt : integer;

 s : string;

{                         Описания подпрограмм                      }

{1. Описание функции koln                                           }

{   Функция определения количества отрицательных элементов в матрице}

function koln(a:matrix; m:integer):integer;

var

 i,j,mx : integer;

begin

 mx:=0;

 for i:=1 to m do

  for j:=1 to m do

   if a[i][j]<0 then mx:=mx+1;

 koln:=mx;

end;

{ 2. Описание процедуры умножения матриц  umn(a,b,m,r). r=a*b       }
procedure  umn(a, b:matrix; m:integer; var r:matrix);

var

 i,j,k:integer;

 z:real;

begin

 for i:=1 to m do

  for j:=1 to m do

   begin

    z:=0;

    for k:=1 to m do

     z:=z+ a[i,k]*b[k,j];

    r[i,j]:=z;

   end;

end;

{ 3. Описание процедуры trans.   r=b транспонированная               }

procedure trans( b:matrix; m:integer; var r:matrix);

var

 i,j:integer;

begin

 for i:=1 to m do

  for j:=1 to m do

   r[i,j]:=b[j,i];

end;

{ 4. Описание процедуры input – ввод квадратной матрицы  A[m,m]      }

procedure  input( var a:matrix; m:integer);

var

 i,j:integer;

begin

clrscr;

 writeln('  ввод матрицы  порядка ',m);

writeln(' Ввод каждого элемента завершайте нажатием EnteR');

  for i:=1 to m do

  begin

   gotoxy(1,i+2);    write('  row #',i);

   for j:=1 to m do

   begin

    gotoxy(j*6+8,i+2); readln(a[i,j])

  end;

  end ;

end;

{ 5. Описание процедуры print. Вывод квадратной матрицы A[m,m]      }

procedure print( a:matrix; m:integer);

var

 i,j:integer;

begin

 for i:=1 to m do

 begin

  for j:=1 to m do

   write(a[i,j]:8:2);

  writeln

 end;

end ;

begin

 {                       главный  модуль                            }

 clrscr;

 {ввод порядка матриц с контролем}

 repeat

  write('Read row  '); readln(row);

 until (row>0)and(row<=n);

 writeln('  ввод матрицы  С  ');

 input (c,row);

 writeln('  ввод матрицы   T  ');

 input (t,row);

 writeln('  Матрица С') ;

 print(C,row);

 kc:= koln(c,row);

 writeln('count of negative in C ',kc);

 writeln('  Матрица T') ;

 print(T,row);

 kt:= koln(t,row);

 writeln('count of negative in T ',kt);

 if kc>kt then

        begin

         { Вычисление матрицы F }

          umn(c, t,row,r);

          trans(r, row,f);

          s:='=(c*t) transp';

         end else

         begin

          { f  = c транспонированная, умножена на t }

            s:='=C транспонированная, умножена на t ';

            trans(c, row,r);

            umn(r, t,row,f);

            { Вычисленa матрицa F}

          end;

 writeln('   Матрица  F'+s);

 print(f,row);

 readln;

end.

2.5.2. Задания к лабораторной работе 

Цель работы:

· получение навыков нисходящего проектирования, структурного программирования, тестирования, оформления программы;

· применение процедур и функций.

Порядок выполнения работы

1. Внешнее проектирование программы:

· определение входных и выходных данных;

· определение модульной структуры программы;

· проектирование тестовых наборов.

Результаты внешнего проектирования представляются в виде следующих документов:

· схема иерархии модулей;

· таблица спецификации модулей;

· алгоритм главного модуля;

· таблица сообщений;

· формат подготовки к вводу исходных данных;

· формы выдачи результатов;

· таблицы тестовых данных (по принципу «черного ящика»).

2. Внутреннее проектирование программы:

Для каждого модуля разработать: структуру и состав его внутренних данных; алгоритм работы; дополнительные тесты для проверки работы модуля.

3. Контроль алгоритма, соответствие спецификациям.

4. Кодирование и отладка алгоритма программы.

5. Тестирование отлаженной программы.

6. Оформление и защита отчета.

Требования к оформлению отчета

Следует помнить: документы, входящие в отчет, предназначаются для чтения пользователям, а не автору программы.
1.Схема иерархии модулей должна состоять из графического изображения связей модулей по управлению и текста, описывающего назначение модуля (рис. 12).

Рис. 12. Пример описания схемы иерархии модулей
2. Таблица спецификаций составляется для каждого модуля, содержит информацию обо всех входных и выходных данных:

· имя данных, используемое в спецификациях;

· описание алгоритма, блок-схема к программе;

· назначение данных;

· допустимые диапазоны значений;

· единицы измерений, тип величины (не использовать служебные слова языка программирования).

3. Таблица сообщений об ошибках должна содержать следующую информацию:

· имя модуля, печатающего сообщение;

· полный и точный текст печатаемого сообщения;

· описание причины, вызвавшей печать сообщения;

· рекомендации пользователю по организации поиска и устранения ошибки.

4.Инструкции по подготовке исходных данных должны содержать исчерпывающее описание ввода данных (порядок следования, разделители).

5. Макет выходного документа должен давать полное представление о виде, размерах и содержимом печатаемого документа.

6. Таблица тестовых данных содержит следующую информацию:

· какая часть или особенность алгоритма программы тестируется;

· состав входных данных;

· ожидаемые результаты.

7. Таблица основных структур данных должна содержать исчерпывающую информацию обо всех основных (упоминаемых в описании алгоритма или блок-схеме, но не включенных в таблицу внешних спецификаций) структурах данных, то есть структуру, тип, размерность, атрибуты, семантику переменных.

8. Краткое словесное описание алгоритма программы.

Требования к оформлению программы

1. В начале программы должен быть помещен комментарий вида:

{             название программы, назначение программы

                     автор программы (ф.и.о., группа)

          дата написания или последней корректировки         }

2. В начало каждого модуля должен быть помещен комментарий, описывающий функции данного модуля.

3. В тексте программы в ключевых точках  должны быть помещены комментарии, текст которых сообщает об особенностях алгоритма и по возможности соответствует тексту из блок-схемы.

Индивидуальные задания на процедуры и функции
Замечания:
· Обмен данными между модулями осуществлять только через список параметров. 

· Структурные типы формальных параметров необходимо описать в разделе типов главного модуля.

· Если в задаче используются несколько массивов, отличающихся только границами своих измерений, целесообразно объявить глобальный тип с максимальными значениями по каждой размерности. Имя этого типа будет типом формального параметра в заголовке модуля.
13. Даны массивы Х[15], Z[20,15]

2.5.3. Контрольные вопросы и задания
1. Какова структура описания процедуры и функции?

2. В чем состоит отличие процедуры от функции?

3. Что такое область действия идентификаторов?

4. Каковы основные правила определения области действия для идентификаторов процедур и функций?

5. Какие параметры называются формальными, а какие фактическими?

6. По каким признакам различаются параметры?

7. Какие способы передачи параметров реализованы в Паскале?

8. Каковы правила передачи параметров-значений?

9. Каковы правила передачи параметров-ссылок? 

10. В чем особенности бестиповых параметров?

11. Для чего предназначены директивы near и far?

12. В чем особенности использования директивы forward?

13. Напишите подпрограмму поиска максимального элемента в массиве.

14. Напишите подпрограмму подсчета количества положительных элементов квадратной матрицы порядка N.

CURS – головной модуль.


INPUT – модуль ввода данных.


CHECK – модуль проверки области допустимых значений данных.


PROCDT – модуль обработки данных.


PRTOUT – модуль, печатающий выходной документ.

